

Modelo Fasorial para Sistemas Eólicos Baseados no DFIG em Modelica

José Victor M. Uemura* Marcelo A. Tomim**

* Universidade Federal de Juiz de Fora, MG, (e-mail: mauad.uemura@engenharia.ufjf.br).

** Universidade Federal de Juiz de Fora, MG, (e-mail: marcelo.tomim@ufjf.br).

Abstract: This paper aims at presenting the implementation of a phasor model for wind conversion systems based on the doubly-fed induction generator using the Modelica language. The model consists of several subsystems, which will be detailed along with justifications for the adopted simplifications. The control systems for this configuration are implemented using vector control, which allows the voltage-source converter to operate as a current-controlled source. Reference values for the controllers are based on the control regions of a variable-speed wind turbine, which will be verified through the simulation of the implemented model.

Resumo: O objetivo do presente trabalho é apresentar a implementação, através da linguagem Modelica, de um modelo fasorial para sistemas eólicos baseados no gerador de indução duplamente alimentado. O modelo é composto por diversos subsistemas, que serão detalhados com justificativas para as simplificações adotadas. A implementação dos sistemas de controle dessa configuração é realizada através do controle vetorial, que permite o conversor operar como uma fonte de corrente controlada. As referências para os controladores são baseadas nas regiões operativas de uma turbina eólica de velocidade variável, que serão verificadas através da simulação do modelo implementado.

Keywords: REDs; phasor model; DFIG; control; Modelica.

Palavras-chaves: REDs; modelo fasorial; DFIG; controle; Modelica;

GLOSSÁRIO

CA Corrente alternada
CC Corrente contínua
DFIG *Doubly-Fed Induction Generator*
FMI *Functional Mock-up Interface*
MPPT *Maximum Power Point Tracking*
OMEdit *OpenModelica Connection Editor*
PI Proporcional-integral
PLL *Phase-lock Loop*
PWM *Pulse Width Modulation*
RED Recurso Energético Distribuído
WECS *Wind Energy Conversion System*

1. INTRODUÇÃO

Nos últimos anos, os recursos energéticos distribuídos (RED), que incluem fontes renováveis como solar, eólica e biomassa, bem como sistemas de armazenamento de energia, têm se mostrado uma alternativa cada vez mais viável e atraente para o setor elétrico brasileiro e global (EPE, 2019). Embora os REDs ofereçam inúmeras vantagens, como a redução de custos de energia e a diminuição na emissão de poluentes, operadores e outros agentes têm se preocupado com a caracterização adequada dos equipamentos que compõem os REDs, especialmente

com o aumento da participação desses recursos na matriz energética.

Dentro da ampla variedade de REDs, encontram-se os sistemas de geração eólica (WECS), que podem apresentar configurações distintas a depender do tipo de aplicação desejada. Os diferentes tipos de sistemas de conversão eólica são subdivididos através de sua capacidade de variação de velocidade, tipo de gerador, entre outros fatores como o tipo de conversor e presença de caixa de engrenagens. Merecem destaque cinco principais topologias frequentemente difundidas na literatura (Yaramasu et al., 2015):

- Tipo 1: velocidade fixa ($\pm 1\%$) com gerador assíncrono tipo gaiola e caixa de engrenagens;
- Tipo 2: velocidade semi-variável ($\pm 10\%$) com gerador assíncrono de rotor bobinado e caixa de engrenagens;
- Tipo 3: velocidade semi-variável ($\pm 30\%$), conversor de potência parcialmente dimensionado, gerador de indução duplamente alimentado (DFIG) e caixa de engrenagens;
- Tipo 4: velocidade totalmente variável (0 – 100%), conversor de potência com capacidade plena e gerador assíncrono tipo gaiola ou síncrono com ímãs permanentes;
- Tipo 5: velocidade totalmente variável (0–100%) com conversor de conjugado/velocidade, gerador síncrono de rotor bobinado e caixa de engrenagens.

* Agradecimentos especiais ao suporte financeiro da UFJF, Petrobrás, CAPES, CNPq, FAPEMIG e INERGE.




Figura 1. Diagrama do modelo construído através do OMEdit para sistemas eólicos baseados no DFIG.

Dentre as topologias listadas, os sistemas de conversão do tipo 3 são atualmente os mais empregados, com uma participação de cerca de 50% do mercado mundial (Cardenas et al., 2013). Por outro lado, nos empreendimentos em alto mar, existe a tendência de emprego de turbinas de maior capacidade e tamanho, com necessidade de manutenção reduzida, favorecendo, assim, os sistemas do tipo 4.

Visando aprofundamento na modelagem dos REDs, o presente artigo se concentra na modelagem dos sistemas de conversão do tipo 3, como ilustrado na Figura 1. Nesse sistema os enrolamentos do estator e do rotor são alimentados pela rede e por um conversor eletrônico de potência (*back-to-back*), respectivamente. Como o conversor fornece um caminho bidirecional para o fluxo de energia do rotor, a eficiência dessa configuração é maior quando comparada a configurações mais simples como as do tipo 1 e 2 (Datta e Ranganathan, 2003). Além de melhorar o desempenho dinâmico e permitir o rastreamento do ponto de máxima potência (MPPT), o conversor *back-to-back* aumenta a faixa de variação da velocidade da máquina para até 30% da velocidade síncrona (Liserre et al., 2011).

Com o objetivo de caracterizar adequadamente os sistemas de geração eólica baseados no DFIG para estudos de transitórios eletromecânicos, um modelo de sequência positiva será desenvolvido utilizando-se da linguagem Modelica (The Modelica Association, 2021). Essa linguagem já vem sendo utilizada em trabalhos acadêmicos envolvendo sistemas de potência, tanto para análises de regime permanente (Henriques et al., 2020) como dinâmico (Guiroquet et al., 2018; de Castro et al., 2023).

A linguagem de modelagem matemática Modelica, apresenta vantagens como a possibilidade de se declarar um determinado modelo dinâmico através de equações na forma literal e acausal. Além de contar com diversos recursos como parte da interface gráfica de desenvolvimento, tal como o OMEdit da plataforma de simulação OpenModelica. Os recursos gráficos disponibilizados pelo OMEdit que permitem representações gráficas de modelos e seus subsistemas, assim como suas interconexões, foram empregados no presente desenvolvimento, como ilustrado na Figura 1, que apresenta o sistema eólico em questão.




Figura 2. Diagrama fasorial dos sistemas referenciais.

Algoritmo 1 Modelo da transformada.

```

1  model T
2  Modelica.Blocks.Interfaces.RealInput u[2];
3  Modelica.Blocks.Interfaces.RealOutput y[2];
4  Modelica.Blocks.Interfaces.RealInput delta;
5  equation
6  y[1] = cos(delta) * u[1] + sin(delta) * u[2];
7  y[2] = (-sin(delta) * u[1] + cos(delta) * u[2];
8  end T;

```

2. MODELOS DESENVOLVIDOS

O sistema eólico baseado no DFIG é composto essencialmente pelos subsistemas da máquina de indução, conversor e turbina, além dos controles elétrico e mecânico associados. Será apresentado, no entanto, primeiramente, o sistema referencial adotado para os diversos modelos desenvolvidos.

2.1 SISTEMA REFERENCIAL

Para o tratamento de grandezas senoidais trifásicas, típicas na modelagem de máquinas elétricas e conversores de potência, emprega-se, comumente, a transformação de Park, que estabelece uma relação entre grandezas senoidais fixas no espaço e grandezas contínuas girantes no espaço. Essas últimas são representadas em um sistema referencial com eixos ortogonais entre si e que giram no tempo, tipicamente conhecidos como eixos direto (d) e em quadratura (q).

Dado o objetivo de se modelar os sistema de conversão eólica no domínio fasorial, deve-se estabelecer uma relação entre as grandezas referidas ao sistema de coordenadas qd e os fasores no plano complexo. Antes de mais nada, um sinal senoidal $f(t)$ está associado ao seu fasor \bar{F} através da relação abaixo.

$$f(t) = \text{Re} \left\{ \sqrt{2} \bar{F} e^{j\omega t} \right\} \quad (1)$$

Adotando-se agora que o sistema referencial qd gire com rotação idêntica à frequência do sinal elétrico ω , as projeções do sinal $f(t)$ ao longo dos eixos q e d permanecem constantes, a exemplo do fasor \bar{F} . Assim, pode-se estabelecer uma relação entre grandezas no sistema referencial qd e o sistema referencial arbitrário dos fasores a partir da ilustração apresentada na Figura 2. Essa relação é denotada na forma complexa em (2), enquanto sua forma real é descrita no Algoritmo 1.

$$F_q - jF_d = \bar{F} \angle -\delta \quad (2)$$

As entradas do modelo apresentado no Algoritmo 1 consistem num vetor u cujas componentes correspondem às partes real e imaginária de um fasor definido num sistema

Algoritmo 2 Modelo da máquina de indução.

```

1  model MIT
2  import Units=DWT.Units;
3  import MMR=Modelica.Mechanics.Rotational;
4  import CI=DWT.Circuit.Interfaces;
5  parameter SI.time Hm = 0.52 "Constante de Inércia [s]";
6  parameter Units.PerUnit Rs = 0.01 "Resistência do estator [pu]";
7  parameter Units.PerUnit Rr = 0.01 "Resistência do rotor [pu]";
8  parameter Units.PerUnit Lls = 0.1 "Dispersão do estator [pu]";
9  parameter Units.PerUnit Llr = 0.08 "Dispersão do rotor [pu]";
10 parameter Units.PerUnit Lm = 3.0 "Magnetização [pu]";
11 parameter SI.AngularVelocity Wb = 377 "Velocidade angular base
    ↪ [rad/s]"
12 Units.PerUnit Wrm "Velocidade do rotor [pu]";
13 Units.PerUnit Te "Conjugado elétrico [pu]";
14 Units.PerUnit Vqs, Vds "Componentes de tensão no estator [pu]";
15 Units.PerUnit Vqr, Vdr "Componentes de tensão no rotor [pu]";
16 Units.PerUnit Iqs, Ids "Componentes de corrente no estator [pu]";
17 Units.PerUnit Iqr, Idr "Componentes de corrente no rotor [pu]";
18 Units.PerUnit fqs, fds "Componentes de fluxo no estator [pu]";
19 Units.PerUnit fqr, fdr "Componentes de fluxo no rotor [pu]";
20 CI.PositivePin pin_estator "Terminal do estator";
21 CI.NegativePin pin_rotor "Terminal do rotor";
22 MMR.Interfaces.Flange_a eixo "Flange do rotor";
23 MMR.Components.Inertia inertia_maq(J=2*Hm) "Modelo do rotor";
24 MMR.Sources.Torque torque "Conjugado aplicado ao rotor";
25 initial equation
26   der(Wrm) = 0;   inertia_maq.phi = 0;
27   der(fqr) = 0;   der(fdr) = 0;
28 equation
29   Vqs = pin_estator.v.re; Vds = pin_estator.v.im;
30   Iqs = pin_estator.i.re; Ids = pin_estator.i.im;
31   Vqr = pin_rotor.v.re; Vdr = pin_rotor.v.im;
32   Iqr = pin_rotor.i.re; Idr = pin_rotor.i.im;
33   Te = torque.tau;
34   Wrm = inertia_maq.w;
35   fqs = Lls * Iqs + Lm * (Iqs + Iqr);
36   fds = Lls * Ids + Lm * (Ids + Idr);
37   Vqs = rs * Iqs + fds; // ( der(fqs)/Wb ) negligenciada
38   Vds = rs * Ids - fqs; // ( der(fds)/Wb ) negligenciada
39   fqr = Llr * Iqr + Lm * (Iqs + Iqr);
40   fdr = Llr * Idr + Lm * (Ids + Idr);
41   Vqr = rr * Iqr + (1 - Wrm) * fdr + der(fqr)/Wb;
42   Vdr = rr * Idr - (1 - Wrm) * fqr + der(fdr)/Wb;
43   Te = fds * Iqs - fqs * Ids;
44   connect(eixo, inertia_maq.flange_a);
45   connect(inertia_maq.flange_b, torque.flange);
46 end MIT;

```

referencial arbitrário, juntamente do ângulo de transformação δ . A saída do modelo corresponde ao vetor y que representa as grandezas referenciadas ao sistema de coordenadas qd .

Neste trabalho, adota-se para a máquina de indução o referencial em que o eixo em quadratura está alinhado com o eixo real do referencial arbitrário. Já para os controladores dos conversor *back-to-back*, um circuito de sincronismo, conhecido como PLL, é responsável por estimar um ângulo de transformação de forma a alinhar o eixo em quadratura com o fasor de tensão nos terminais do estator.

2.2 MÁQUINA DE INDUÇÃO

O subsistema associado à máquina de indução desenvolvido através da linguagem Modelica é apresentado no Algoritmo 2. Os conectores elétricos fasoriais (**PositivePin** e **NegativePin**) estabelecem as restrições associadas às leis de Kirchhoff para tensões e correntes tanto para o terminal do estator como do rotor. Já o conector mecânico (**Flange_a**) estabelece relações entre a posição angular e a transferência de conjugado no eixo da máquina.

O modelo implementado para a máquina de indução é baseado na modelagem desenvolvida por Krause et al. (2002). De forma a possibilitar a conexão da máquina a uma rede fasorial, considera-se o estator também em regime permanente, eliminando-se as parcelas derivativas




Figura 3. Diagrama do conversor através do OMedit.

Algoritmo 3 Modelo do conversor.

```

1  model CONVERSOR
2  import Units=DWT.Units;
3  import CI=DWT.Circuit;
4  import CPLX=Modelica.ComplexBlocks.Interfaces;
5  import BLK=Modelica.Blocks.Interfaces;
6  import Analog=Modelica.Electrical.Analog;
7  import SI=Modelica.Units.SI;
8  parameter SI.Voltage VCAb = 690. "Tensão base CA (linha)";
9  parameter SI.Voltage VCCb = 1400. "Tensão base CC";
10 parameter Units.PerUnit Ceq = 35.897 "Capacitância [pu]";
11 CI.Sources.ControlledVoltageSource VSR "Tensão controlada (rotor)";
12 CI.Sources.ControlledVoltageSource VSG "Tensão controlada (rede)";
13 CPLX.ComplexInput Mr "Fator de modulação (rotor)";
14 CPLX.ComplexInput Mg "Fator de modulação (rede)";
15 BLK.RealOutput outVcc "Medição da tensão no capacitor";
16 CI.Interfaces.PositivePin outR "Terminal (rotor)";
17 CI.Interfaces.PositivePin outG "Terminal (rede)";
18 Analog.Basic.Ground ground "Nó de referência";
19 Analog.Basic.Capacitor capacitor(C = Ceq) "Capacitor do elo CC";
20 Analog.Sources.SignalCurrent IccRSC "Corrente controlada (rotor)";
21 Analog.Sources.SignalCurrent IccGSC "Corrente controlada (rede)";
22 protected
23   parameter Units.PerUnit Kc = sqrt(3/2)/2*VCCb/VCAb "Relação de
    ↪ transformação (PWM)";
24 initial equation
25   der(capacitor.v) = 0;
26 equation
27   VSR.S.re = IccRSC.i*capacitor.v;
28   VSG.S.re = IccGSC.i*capacitor.v;
29   VSR.v = Kc * capacitor.v * Mr;
30   VSG.v = Kc * capacitor.v * Mg;
31 end CONVERSOR;

```

dos fluxos enlaçados pelo estator, tal como apresentado nas linhas 37 e 38 do Algoritmo 2. Essas parcelas derivativas caracterizam o efeito transformador da máquina que pouco influenciam sua dinâmica no período em que as interações eletromecânicas são mais pronunciadas.

2.3 CONVERSOR DE POTÊNCIA

A implementação do conversor de potência é apresentada através do Algoritmo 3 e da Figura 3. Em linhas gerais, o conversor é realizado por meio de fontes controladas de tensão CA, acopladas ao lado CC através de fontes de corrente também controladas. Os sinais de referência para essas fontes são definidos implicitamente através do balanço de potência apresentado nas linhas 27 e 28 do Algoritmo 3. As perdas por condução e chaveamento são, por simplicidade, ignoradas.

Os sinais de referência das fontes CA controladas são obtidos através dos fatores de modulação, M_r e M_g , como apresentado nas linhas 29 e 30 do Algoritmo 3, que estão relacionados à lógica de chaveamento PWM (Iravani e Yazdani, 2010). Nesse contexto, é fundamental levantar a relação dos fatores de modulação com as tensões terminais do conversor como apresentado abaixo.

$$\bar{V}_r = (K_C V_{CC}) \bar{m}_r \quad \bar{V}_g = (K_C V_{CC}) \bar{m}_g \quad (3)$$

A constante K_C , definida em (4), serve a dois objetivos. Primeiro, K_C compatibiliza os sistemas por unidade das partes CA e CC do conversor. Segundo, K_C também

Algoritmo 4 Função do coeficiente de potência.

```

1 function CP
2   input Real Lambda "Velocidade específica da turbina";
3   input Real Beta "Ângulo de inclinação das pás";
4   output Real Cp "Coeficiente de potência";
5   protected
6     Real Alpha;
7     constant Real C1 = 0.22;
8     constant Real C2 = 116;
9     constant Real C3 = 0.4;
10    constant Real C4 = 0;
11    constant Real C5 = 0;
12    constant Real C6 = 5;
13    constant Real C7 = 12.5;
14    constant Real C8 = 0.08;
15    constant Real C9 = 0.035;
16  algorithm
17    Alpha := 1/(1/(Lambda + Beta*C8) - C9/(Beta^3 + 1));
18    Cp := C1*(C2/Alpha - C3*Beta - C4*Beta^C5 - C6)*exp(-C7/Alpha);
19  end CP;

```

introduz o efeito da lógica de chaveamento PWM para a produção de tensões CA.

$$K_C = \frac{1}{2} \sqrt{\frac{3}{2}} \frac{V_{CC}^{base}}{V_{CA}^{base}} \quad (4)$$

2.4 TURBINA EÓLICA

O modelo da turbina eólica pode ser dividido em duas partes: modelo aerodinâmico e mecânico.

A equação que caracteriza o modelo aerodinâmico em linguagem Modelica é apresentada na linha 35 do Algoritmo 5, tipicamente encontrada na literatura (Abad et al., 2011; Wu, 2006). Essa expressão considera a velocidade do vento uniforme nas pás da turbina. A função associada ao coeficiente de potência $C_p(\lambda, \beta)$, caracterizado no Algoritmo 4, varia tipicamente entre 25 e 45% (Anaya-Lara et al., 2014) sendo limitada a no máximo 59,3%, ponto esse definido como limite de Betz. Conforme se observa no Algoritmo 4, há uma forte dependência desse coeficiente com a velocidade específica (λ) e o ângulo de inclinação (β). Além disso, para a caracterização apropriada do coeficiente de potência é necessário o ajuste de nove constantes com base em dados de operação ou do próprio fabricante.

Para a representação do sistema mecânico do aerogerador, é comumente adotado um sistema de duas massas acopladas por um eixo e uma caixa de engrenagens. Por simplicidade, o sistema mecânico foi referido ao lado de alta rotação do gerador. A massa associada à turbina é modelada independentemente da massa do gerador (linha 25 do Algoritmo 5), assim como o eixo, caracterizado pelo modelo SpringDamper (linha 27 do Algoritmo 5).

3. SISTEMAS DE CONTROLE

Os sistemas de controle dessa configuração de aerogerador, são estruturados através de cinco principais subsistemas, conforme apresentado na Figura 4. Nesta seção, serão apresentados cada um desses subsistemas e suas principais funções. Esses subsistemas são responsáveis por gerar os fatores de modulação para o conversor *back-to-back*, bem como a referência associada ao ângulo mecânico de inclinação das pás da turbina eólica.

3.1 OBSERVADOR DE ESTADOS

Tipicamente, é necessário estimar os estados do sistema para realizar as ações de controle. Nesse caso, o PLL é

Algoritmo 5 Modelo da turbina eólica.

```

1 model TURBINA
2   import SI = Modelica.Units.SI;
3   import pi = Modelica.Constants.pi;
4   import Units = DWT.Units;
5   import MMR = Modelica.Mechanics.Rotational;
6   import BLK = Modelica.Blocks;
7   parameter SI.Power Pb = 2e6 "Potência base [W]";
8   parameter SI.AngularVelocity Wrmb = 60*pi "Frequência base [rad/s]";
9   parameter SI.Time Ht = 4.2 "Constante de Inércia [s]";
10  parameter SI.Density par = 1.225 "Densidade do ar [kg/m³]";
11  parameter SI.Length R = 37.5 "Raio das pás [m]";
12  parameter Real N = 111.5 "Relação de engrenagens";
13  parameter Units.PerUnit Ktm = 0.3 "Const. de elasticidade do eixo";
14  parameter Units.PerUnit Dtm = 1.5 "Const. de amortecimento do eixo";
15  parameter Real kb = 0.7143 "Ganho do servomotor";
16  parameter Real tb = 1.1905 "Constante de tempo do servomotor";
17  Units.PerUnit Ttur "Conjugado desenvolvido [pu]";
18  Units.PerUnit Wtur "Velocidade da turbina [pu]";
19  Units.PerUnit Ptur "Potência desenvolvida [pu]";
20  Real cp "Coeficiente de potência";
21  MMR.Interfaces.Flange_a flange_Eixo "Conector para o eixo";
22  BLK.Interfaces.RealInput Vw "Velocidade do vento";
23  BLK.Interfaces.RealInput Beta "Ângulo de inclinação";
24  BLK.Continuous.TransferFunction tfBeta(a = {tb, 1, kb}, b = {kb},
    ← initType = BLK.Types.Init.SteadyState) "Servomotor (Beta)";
25  MMR.Components.Inertia inertia_tur(J = 2*Ht) "Inércia (turbina)";
26  MMR.Sources.Torque torque "Conjugado desenvolvido";
27  MMR.Components.SpringDamper eixo(a_rel(c = Ktm, d = Dtm) "Eixo";
28  initial equation
29    der(Wtur) = 0;
30    der(eixo.phi_rel) = 0;
31  equation
32    Ttur = torque.tau;
33    Wtur = inertia_tur.w;
34    cp = CP(R*(Wtur*Wrmb/N)/Vw, tfBeta.y);
35    Ptur = 0.5*(par*pi*R^2*cp*Vw^3)/Pb;
36    Ttur = Ptur/Wtur;
37    connect(Beta, tfBeta.u);
38    connect(torque.flange, inertia_tur.flange_b);
39    connect(inertia_tur.flange_a, eixo.flange_a);
40    connect(eixo.flange_b, flange_Eixo);
41  end TURBINA;

```




Figura 4. Diagrama causal para os sistemas de controle através do OMEdit.

utilizado como parte do subsistema SYS/MED. O objetivo do PLL é alinhar o eixo q do sistema de referência qd com o fasor da tensão de fase terminal do estator. Esse procedimento simplifica o controle do sistema e permite o desacoplamento das malhas de potência ativa e reativa.

A malha do PLL foi implementada empregando-se a transformação apresentada no Algoritmo 1 junto a um controlador PI. O controlador PI visa anular seu sinal de entrada, que corresponde à componente V_d do fasor observado (\bar{V}_t), através do ajuste do ângulo δ da transformação. Desta forma, o ângulo δ de regime permanente consiste na diferença angular entre a referência arbitrária da rede e a tensão terminal da máquina.

Com as variáveis estimadas pelo PLL, pode-se determinar as correntes associadas tanto ao rotor quanto à rede

elétrica no sistema referencial de controle, fundamentais para a modelagem dos subsistemas subsequentes.

3.2 TABELAS DE PESQUISA

A operação dos sistemas de conversão eólica baseados no DFIG seguem tipicamente características de potência e velocidade da máquina em função da velocidade do vento. Essas características permitem dividir a operação da turbina em quatro regiões distintas. Enquanto na região 1, objetiva-se manter a velocidade da turbina em um valor mínimo, nas regiões 3 e 4, a velocidade da turbina deve ser mantida em seu valor nominal. Na região 2, o controle visa converter a máxima potência do vento mantendo-se o coeficiente de potência C_p em seu valor máximo através do ajuste da velocidade da turbina. Apesar da velocidade ser mantida nominal nas regiões 3 e 4, nessa última, a potência também deve ser regulada em seu valor nominal. Para a realização dessa ação de controle, aumenta-se o ângulo β de inclinação das pás na medida em que o vento aumenta, reduzindo, assim, a potência convertida através da redução do coeficiente de potência C_p .

Com o intuito de automatizar a geração de tabelas com as referências de velocidade da turbina (ω_{rm}) e do ângulo de inclinação das pás (β), foi desenvolvido um algoritmo em *Python* descrito no arquivo `controlWECS.ipynb` (listado no repositório, Apêndice A). As tabelas geradas seguem o formato exigido pelo interpolador implementado através do bloco `CombiTable1Ds`, contida na biblioteca padrão do Modelica.

3.3 CONTROLE VETORIAL

No presente trabalho, será utilizada a estratégia de controle vetorial que torna o conversor uma fonte de corrente controlada.

Os controles de velocidade mecânica do eixo, potência reativa nos terminais do aerogerador e tensão no barramento CC, são realizados através das correntes sintetizadas pelo conversor. Nesse contexto, faz-se necessário modelar apropriadamente as malhas de controle do subsistema REF/CUR. Esses desenvolvimentos, no entanto, são detalhados em trabalhos anteriores dos autores [Uemura e Tomim \(2022\)](#), que estabelecem as relações entre as variáveis a serem controladas e as correntes do conversor, bem como a justificativa para o emprego de controladores PI para as malhas de controle em questão.

3.4 CONTROLE DE CORRENTES

Tendo em vista que o conversor de potência sintetiza tensões através de lógicas de chaveamento, tais como a lógica PWM, é necessário estabelecer, em princípio, relações entre tensões e correntes nos terminais do conversor.

Para que o conversor opere como uma fonte de corrente, faz-se necessário implementar malhas de controle que estabeleçam tensões a serem sintetizadas a partir de valores de referência para as correntes. Essas malhas estão contidas no subsistema CUR/VOL.

De forma a superar as dinâmicas de primeira ordem que relacionam as tensões e correntes do rotor, é utilizado um




Figura 5. Resultados da simulação do modelo desenvolvido através da linguagem Modelica.

compensador PI. O mesmo princípio se aplica ao conversor associado à rede elétrica, que normalmente utiliza um filtro RL nas interfaces conversor-rede. Esse filtro incorre em uma dinâmica de primeira ordem dominante entre tensões e correntes, assim como no circuito do rotor.

3.5 FATORES DE MODULAÇÃO

Conforme apresentado na relação (3), os fatores de modulação representam relações de transformação média de correntes das lógicas de chaveamento implementadas no conversor de forma que as tensões CA sejam sintetizadas a partir da tensão disponível no barramento CC. Com isso em mente, o subsistema VOL/MOD é dedicado a obter os fatores de modulação a partir das componentes da referência de tensão geradas pelas malhas de controle de corrente.

Ademais, deve-se atentar para o fato de que as lógicas de controle são definidas no sistema referencial qd do conversor, no qual a tensão do estator da máquina é alinhada ao eixo em quadratura. Dessa forma, ainda dentro do subsistema VOL/MOD, emprega-se o ângulo estimado pelo PLL para se referir os fasores calculados pelo sistema de controle ao sistema referencial arbitrário da rede elétrica.

4. SIMULAÇÃO DO SISTEMA

A simulação para a verificação do modelo fasorial do sistema eólico baseado no DFIG, desenvolvido através da linguagem Modelica, foi realizada variando-se linearmente a velocidade do vento de 4 até 25 m/s durante 1000s, passando por todas as regiões operativas da turbina eólica. A Figura 5 apresenta os resultados obtidos para esta simulação.

Através destes resultados evidencia-se o funcionamento esperado das malhas de controle associadas. Por exemplo, pode-se observar que o sistema de controle foi capaz de

manter a velocidade constante ao longo das regiões 1 (velocidade mínima de turbina) e 3 (velocidade máxima de turbina com potência inferior à nominal). Adicionalmente, a máxima potência também foi rastreada na faixa de velocidade de vento associada à região 2, dado que o coeficiente de potência é mantido constante em seu valor máximo. Finalmente, para velocidades de vento superiores a nominal de 12 m/s, observa-se o decaimento do coeficiente de potência através do aumento do ângulo β de inclinação da pá, de forma que tanto a potência nominal como a velocidade da turbina sejam mantidas constantes.

5. CONCLUSÃO

No presente trabalho, pôde-se apresentar o modelo fasorial de sequência positiva, desenvolvido para sistemas de conversão de energia eólica baseados no gerador de indução duplamente alimentado (DFIG), através da linguagem Modelica. Resultados foram apresentados para uma simulação do sistema, frente uma variação em rampa da velocidade do vento, demonstrando que o comportamento dinâmico do aerogerador atende aos requisitos de controle pré-estabelecidos.

É importante também salientar que a linguagem Modelica permite a representação de modelos elétricos e mecânicos via equações, tanto algébricas como dinâmicas, enquanto controladores podem ser convenientemente caracterizados por seus diagramas de blocos associados. Além disso, por ser orientada a objetos, a linguagem Modelica permite o desenvolvimento de um modelo portátil e reutilizável, podendo, inclusive ser compilado em uma biblioteca de acesso dinâmico que atenda à interface padronizada FMI. Essa característica favorece a separação do desenvolvimento de modelos das ferramentas de simulação no tempo, que agora podem ser integradas dentro de um contexto de cossimulação. Estudos vêm sendo realizados nesse sentidos.

O modelo completo se encontra disponível no repositório descrito no Apêndice A, juntamente aos arquivos em linguagem *Python* que auxiliam no projeto e sintonia dos controladores que integram o sistema de conversão eólica presentemente analisado.

AGRADECIMENTOS

Os autores gostariam de agradecer à PETROBRAS, à Coordenação de Aperfeiçoamento de Pessoal de Nível Superior (CAPES), ao Conselho Nacional de Desenvolvimento Científico e Tecnológico (CNPq), à Fundação de Amparo à Pesquisa do Estado de Minas Gerais (FAPEMIG) e ao Instituto Nacional de Energia Elétrica (INERGE). Esse trabalho foi desenvolvido em parceria com a PETROBRAS dentro do projeto intitulado "Gerenciamento de Recursos Energéticos Distribuídos para Prestação de Novos Serviços à Rede Elétrica", PD-00553-0064/2019, no âmbito do programa de P&D do sistema elétrico regulado pela ANEEL.

REFERÊNCIAS

Abad, G., Lopez, J., Rodriguez, M., Marroyo, L., e Iwanski, G. (2011). *Doubly Fed Induction Machine: Modeling and Control for Wind Energy Generation*. John Wiley & Sons.

- Anaya-Lara, O., Campos-Gaona, D., Moreno-Goytia, E., e Adam, G. (2014). *Offshore Wind Energy Generation - Control, Protection, and Integration to Electrical Systems*. Wiley-IEEE Press, New York, USA, 2 edição.
- Cardenas, R., Peña, R., Alepuz, S., e Asher, G. (2013). Overview of Control Systems for the Operation of DFIGs in Wind Energy Applications. *IEEE Transactions on Industrial Electronics*, 60(7), 2776–2798.
- Datta, R. e Ranganathan, V.T. (2003). A Method of Tracking the Peak Power Points for a Variable Speed Wind Energy Conversion System. *IEEE Transactions on Energy Conversion*, 18(1), 163–168.
- de Castro, M., Winkler, D., Laera, G., Vanfretti, L., Dorado-Rojas, S.A., Rabuzin, T., Mukherjee, B., e Navarro, M. (2023). Version [OpenIPSL 2.0.0] - [iTesla Power Systems Library (iPSL): A Modelica library for phasor time-domain simulations]. *SoftwareX*, 21, 101277. doi:<https://doi.org/10.1016/j.softx.2022.101277>. URL <https://www.sciencedirect.com/science/article/pii/S2352711022001959>.
- EPE (2019). Recursos Energéticos Distribuídos - (PNE 2050). <https://www.epe.gov.br/>. Acessado: 2021-11-27.
- Guironnet, A., Saugier, M., Petitrenaud, S., Xavier, F., e Panciatici, P. (2018). Towards an open-source solution using modelica for time-domain simulation of power systems. Em *2018 IEEE PES Innovative Smart Grid Technologies Conference Europe (ISGT-Europe)*, 1–6. doi:10.1109/ISGTEurope.2018.8571872.
- Henriques, R.M., Tomim, M.A., Machado, M.L., e Passos Filho, J.A. (2020). Implementação de uma biblioteca para a solução do fluxo de potência através da linguagem de modelagem matemática modelica. Em *Congresso Brasileiro de Automática*.
- Iravani, R. e Yazdani, A. (2010). *Voltage-Sourced Converters in Power Systems: Modeling, Control, and Applications*. Wiley-IEEE Press.
- Krause, P.C., Wasynczuk, O., e Sudhoff, S.D. (2002). *Analysis of Electric Machinery and Drive Systems*. Wiley-IEEE Press, New York, USA, 2 edição.
- Liserre, M., Cardenas, R., Molinas, M., e Rodriguez, J. (2011). Overview of Multi-MW Wind Turbines and Wind Parks. *IEEE Transactions on Industrial Electronics*, 58(4), 1081–1095.
- The Modelica Association (2021). Modelica Language. URL <https://www.modelica.org/>.
- Uemura, J.V.M. e Tomim, M.A. (2022). Modelagem e simulação do controle de um gerador de indução duplamente alimentado. Em *XIV Congresso latino-americano de geração e transmissão de energia elétrica - CLAGTEE*, B-8.2-10. doi:10.29327/547386.
- Wu, F. (2006). *Modeling and control of wind turbine with doubly fed induction generator*. p. 1404–1409.
- Yaramasu, V., Wu, B., Sen, P.C., Kouro, S., e Narimani, M. (2015). High-Power Wind Energy Conversion Systems: State-of-the-Art and Emerging Technologies. *Proceedings of the IEEE*, 103(5), 740–788.

Apêndice A. ALGORITMOS DESENVOLVIDOS

Os algoritmos desenvolvidos no escopo do presente trabalho estão disponíveis no repositório do GitHub.

<https://github.com/JVUemura/Uemura2023SBSE>